

# 日々是Oracle APEX

Oracle APEXを使った作業をしていて、気の付いたところを忘れないようにメモをとります。

2022年12月6日 火曜日


## 簡単なファイル管理アプリケーションの作成(2) - 直リンクによるダウンロード

ファイルのIDをURLの引数に与え、指定したファイルを直接ダウンロードできるようにします。

直リンクの実装の詳細、特にページの保護に関する説明については記事「[特定のページへの直リンクによるアクセス](#)」を参照してください。ページ・プロパティのディープ・リンクやページ保護の設定の意味について説明をしています。

本記事では、主に実装の手順について紹介します。

以下のような機能を、本記事では実装します。ダウンロードURLを電子メールなどに含めることにより、ファイルを共有することができます。



## アプリケーション・アイテムの作成


ダウンロードの対象とするファイルを、URLの引数IDによって特定できるようにします。IDには表 SFM\_CONTENTS の列IDの値（つまり主キーの値）を指定します。

共有コンポーネントの **アプリケーション・アイテム** を開きます。




作成済みのアプリケーション・アイテムが一覧されます。

作成を実行します。



作成するアプリケーション・アイテムは名前をIDとし、セッション・ステート保護は制限なしに設定します。設定した後、アプリケーション・アイテムの作成を実行します。



直リンクとなるURLを、セッションの開始時にアプリケーション・アイテムとして設定します。アプリケーションがインストールされているワークスペース名やアプリケーション別名によってURLは変わります。

アプリケーション・アイテムの名前はG\_DOWNLOAD\_URLとします。セキュリティのセッション・ステート保護は制限付き - ブラウザから設定不可を選択し（これがデフォルトです）、アプリケーションのユーザーによる値の設定を禁止します。



アプリケーション・アイテムIDおよびG\_DOWNLOAD\_URLが作成されます。



アプリケーション・アイテムG\_DOWNLOAD\_URLに値を設定します。

共有コンポーネントのアプリケーションの計算を開きます。



作成済みのアプリケーションの計算が一覧されます。作成をクリックします。



計算アイテムとしてG\_DOWNLOAD\_URLを選択します。頻度の計算ポイントは新規インスタンス(新規セッション)開始時です。

計算の計算タイプにファンクション本体を選択し、計算として以下のPL/SQLコードを記述します。

```

/*
 * 簡易URLがONのときにAPEX_UTIL.HOST_URL('SCRIPT')が正しい値を
 * 返さないことがある。以下はそのワークアラウンドとしての実装も含む。
 */
declare
    l_download_url varchar2(800);
    l_apex_path     varchar2(400);
    l_pattern       varchar2(100);
    l_alias         varchar2(200);
begin
    /* https://ホスト名:ポート番号/ords まで。 */
    l_apex_path := apex_util.host_url('APEX_PATH');
    /* ORDS別名を取得する。通常はワークスペース名に一致する。 */
    select pattern into l_pattern from user_ords_schemas
    where parsing_schema = sys_context('USERENV','CURRENT_USER');

```

```


/*
 * アプリケーション別名。日本語の対応を行なっているが効果は未確認。
 * 本来はAPEX_URL.HOST_URL('SCRIPT')はここまでの文字列を返す。
 */
l_alias := utl_url.escape(lower(:APP_ALIAS), false, 'AL32UTF8');
/* 全体としてのURL */
l_download_url := l_apex_path || 'r/' || l_pattern || '/' || l_alias || '/download?id=';
return l_download_url;
end;

```

download\_url.sql hosted with ❤ by GitHub

[view raw](#)

計算の作成をクリックします。



アプリケーションの計算が作成されました。

セッションの開始時にアプリケーション・アイテムG\_DOWNLOAD\_URLに直接ダウンロードするためのURLが設定されます。

## ダウンロード機能の実装

プログラムの中からファイルのダウンロードを呼び出すために、[APEX\\_UTIL.GET\\_BLOB\\_FILE\\_SRC](#)を使用します。このAPIの引数p\_item\_nameとして、ダウンロードするファイルが保存されているBLOB列が割り当てられているページ・アイテムを指定します。このようなページ・アイテムは通常、フォームに含まれています。

今回のアプリケーションでは表SFM\_CONTENTSのフォームが、ページ番号3のSfm Contentとして作成されています。フォームに含まれるページ・アイテムP3\_CONTENTをp\_item\_nameに与えることができます。

ページ・アイテムP3\_CONTENTを使って、ダウンロード機能を実装します。アプリケーションにp\_item\_nameとして割り当て可能なページ・アイテムが含まれない場合は、フォームを実装したページを新たに作成する必要があります。

ダウンロード機能を実装するページを作成します。

ページの作成を実行します。



空白ページを選択します。




ページ定義の名前はdownloadとします。ダウンロードURLのターゲットはこのページ名になります。ページ・モードは標準、ナビゲーションのブレッドクラムの使用、ナビゲーションの使用はともにOFFです。この空白のページがユーザーに表示されることはありません。

ページの作成を実行します。



ページが作成されます。

ページ・プロパティのセキュリティに含まれるディープ・リンクを有効にします。また、ページ・アクセス保護は制限なしにします。



レンダリング前のヘッダー前にプロセスを作成します。


識別の名前はダウンロード、タイプはコードの実行を選択します。PL/SQLコードとして以下を記述します。

```
declare
  l_exist integer;
begin
  -- IDが存在しなければ、ORA-1403 を発生させる。
  select 1 into l_exist from sfm_contents where id = :ID;
  apex_util.redirect_url(
    apex_util.get_blob_file_src(
      p_item_name => 'P3_CONTENT',
      p_v1 => :ID,
      p_content_disposition => 'attachment'
    )
  );
  apex_application.stop_apex_engine;
end;
```

download.sql hosted with ❤ by GitHub

[view raw](#)

サーバー側の条件のタイプにアイテムはNULLではないを選択し、アイテムにIDを指定します。URLの引数としてIDがなければプロセスは実行しません。



以上で実装は完了です。

直接ファイルをダウンロードするURLを確認します。

対話モード・レポートのソースに以下の列DOWNLOAD\_URLを追加します。


:G\_DOWNLOAD\_URL || id download\_url

ページ番号 2 の対話モード・レポートのソースのSQL問合せを以下のSELECT文に置き換えます。

```
select
  id
, title
, abstract
, sys.dbms_lob.getlength(content) content
, content_filename
, content_mimetype
, content_charset
, content_lastupd
, :G_DOWNLOAD_URL || id download_url
from sfm_contents
```

direct\_link\_select.sql hosted with ❤ by GitHub

[view raw](#)



アプリケーションを実行し、対話モード・レポートに表示する列に**Download Url**を追加します。



表示されたURLを直接アクセスすると、ユーザー認証が要求されます。これはページ**download**が**認証が必要なページ**として設定されているためです。正しいユーザー名とパスワードでサインインを行うとファイルのダウンロードが開始します。ページの**ディープ・リンク**が**有効**でなければ、サインインの後は必ずホーム・ページが開くように制限されるため、ページ**download**が開くことはありません。また、**ページ・アクセス保護**が**制限なし**であるため、引数idの指定が許されています。

続く

Yuji N. 時刻: 22:10

共有

[ウェブ バージョンを表示](#)

自己紹介

**Yuji N.**

日本オラクル株式会社に勤務していて、Oracle APEXのGroundbreaker Advocateを拝命しました。  
こちらの記事につきましては、免責事項の参照をお願いいたします。

[詳細プロフィールを表示](#)

Powered by Blogger.

---